

# Translating Haskell to Isabelle

Paolo Torrini<sup>1</sup> and Christoph Lueth<sup>2</sup> Christian Maeder<sup>2</sup> Till Mossakowski<sup>2</sup>

<sup>1</sup> Verimag, Grenoble, Paolo.Torrini@imag.fr

<sup>2</sup> DFKI Lab Bremen,

{Christoph.Lueth,Christian.Maeder,Till.Mossakowski}@dfki.de

**Abstract.** We present partial translations of Haskell programs to Isabelle that have been implemented as part of the Heterogenous Tool Set. The target logic is Isabelle/HOLCF, and the translation is based on a shallow embedding approach.

## 1 Introduction

Automating the translation from programming languages to the language of a generic prover may provide useful support for the formal development and the verification of programs. It has been argued that functional languages can make the task of proving assertions about programs written in them easier, owing to the relative simplicity of their semantics [Tho92,Tho94]. The idea of translating Haskell programs, came to us, more specifically, from an interest in the use of functional languages for the specification of reactive systems. Haskell is a strongly typed, purely functional language with lazy evaluation, polymorphic types extended with type constructor classes, and a syntax for side effects and pseudo-imperative code based on monadic operators [PJ03]. Several languages based on Haskell have been proposed for application to robotics [PHH99,HCNP03]. In such languages, monadic constructors are extensively used to deal with side-effects. Isabelle is a generic theorem-prover implemented in SML supporting several logics — in particular, Isabelle/HOL is the implementation in Isabelle of classical higher-order logic based on simply typed lambda calculus extended with axiomatic type classes. It provides support for reasoning about programming functions, both in terms of rich libraries and efficient automation. Isabelle/HOLCF [MNvOS99] [Pau94,MNvOS99] is Isabelle/HOL conservatively extended with the Logic of Computable Functions — a formalisation of domain theory.

We have implemented as functions of Hets translations of Haskell to Isabelle/HOLCF following an approach based on shallow embedding, mapping Haskell types to Isabelle ones, therefore taking full advantage of Isabelle built-in type-checking. Hets [Mos05,Mos06,MML07] is an Haskell-based application designed to support heterogeneous specification and the formal development of programs. It has an interface with Isabelle, and relies on Programatica [HHJK04] for parsing and static analysis of Haskell programs. Programatica already includes a translation to Isabelle/HOLCF which, in contrast to ours, is based on an object-level modelling of the type system [HMW05].

The paper is organised as follows: Section 2 discusses the semantic background of the translation, while the subsequent sections are devoted to the translation of types, datatypes, classes and function definitions, respectively. Sect. 7 shows some example proofs, and Sect. 8 concludes the paper.

## 2 Semantic Background of the Translation

We firstly describe the subset of Haskell that we cover. Our translation to Isabelle/HOLCF covers at present Booleans, integers, basic constructors (function, product, list, *maybe*), equality, single-parameter type classes (with some limitations), *case* and *if* expressions, *let* expressions without patterns, mutually recursive data-types and functions. It keeps into account partiality and laziness by following, for the main lines, the denotational semantics of lazy evaluation given in [Win93]. There are several limitations: *Prelude* syntax is covered only partially; list comprehension, *where* expressions and *let* with patterns are not covered; further built-in types and type classes are not covered; imports are not allowed; constructor type classes are not covered at all — and so for monadic types beyond list and *maybe*. Of all these limitations, the only logically deep ones are those related to classes — all the other ones are just a matter of implementation.

For the translation, we have followed the informal description of the operational semantics given in the Haskell report [PJ03], and also consulted the complete static semantics for Haskell 98 [Fax02], as well as the (fragmentary) dynamic semantics [HH92]. However, it should be noted that there is no denotational semantics of Haskell! This also has been confirmed in the answers to a query that one of the authors has sent to the Haskell mailing list. There is a static semantics for Haskell. Hence, our translation to Isabelle/HOLCF can be seen as the first denotational semantics given to a large subset of Haskell 98. This also means that there is no notion of correctness of this translation, because it just *defines* the denotational semantics. Of course, an interesting question is the coincidence of denotational and operational semantics. However, this is beyond the scope of the paper.

## 3 Translation of Types

In Isabelle/HOL types are interpreted as sets (class *type*); functions are total and may not be computable. A non-primitive recursive function may require discharging proof obligations already at the stage of definition — in fact, a specific relation has to be given for a function to be proved total. In Isabelle/HOLCF each type is interpreted as a pointed complete partially ordered set (class *pcpo*) i.e. a set with a partial order which has suprema of  $\omega$ -chains and has a bottom. Isabelle's formalisation, based on axiomatic type classes [Wen05], makes it possible to deal with complete partial orders in quite an abstract way. Functions are generally partial and computability can be expressed in terms of continuity. Recursion can be expressed in terms of least fixed-point operator, and so,

in contrast with Isabelle/HOL, function definition does not depend on proofs. Nevertheless, proving theorems in Isabelle/HOLCF may turn out to be comparatively hard. After being spared the need to discharge proof obligations at the definition stage, one has to bear with assumptions on function continuity throughout the proofs. A standard strategy is then to define as much as possible in Isabelle/HOL, using Isabelle/HOLCF type constructors to lift types only when this is necessary.

The provision of `pcpos`, domains and continuous functions by Isabelle/HOLCF eases the translation of Haskell types and functions a lot. However, special care is needed when trying to understand the Haskell semantics. If one reads the section of the Haskell report dealing with types and classes, one could come to the conclusion that a function space in Haskell can be mapped to the space of the continuous functions in Isabelle/HOLCF — this would correspond to a purely lazy language. However, Haskell is a mixed eager and lazy language, and it provides a function `seq` that enforces eager evaluation. This function is introduced in part 6 of the Haskell report, “Predefined Types and Classes”, in section 6.2. We quote from there:

However, the provision of `seq` has important semantic consequences, because it is available at every type. As a consequence,  $\perp$  is not the same as  $\lambda x \rightarrow \perp$ , since `seq` can be used to distinguish them.

In order to enforce this distinction, each function space needs to be lifted. The same holds for products. We define these liftings in a specific Isabelle theory *HsHOLCF* (included in the *Hets* distribution) as follows:

```
defaultsort pcpo

domain ('a,'b) lprod = lpair (lazy lfst :: 'a) (lazy lsnd :: 'b)

domain 'a Lift = Lift (lazy 'a)

types ('a, 'b) "-->" = "('a -> 'b) Lift" (infixr 0)

constdefs
  liftedApp :: "('a --> 'b) => ('a => 'b)" ("_$$$_" [999,1000] 999)
    (* application *)
  "liftedApp f x == case f of
    Lift $ g => g $ x"

constdefs
  liftedLam :: "('a => 'b) => ('a --> 'b)" (binder "Lam " 10)
    (* abstraction *)
  "liftedLam f == Lift $ (LAM x . f x)"
```

Our translation of Haskell types to Isabelle types is defined recursively. It is based on a translation of names for avoidance of name clashes that is not specified here. We write  $\alpha'$  for both the recursive translation of item  $\alpha$  and the

renaming according to the name translation. The translation of types is given by the following rules:

Types

$a$	$\Longrightarrow$	$'a :: \{pcpo\}$
$()$	$\Longrightarrow$	$unit \text{ lift}$
$Bool$	$\Longrightarrow$	$bool \text{ lift}$
$Integer$	$\Longrightarrow$	$int \text{ lift}$
$\tau_1 \rightarrow \tau_2$	$\Longrightarrow$	$\tau'_1 \dashrightarrow \tau'_2$
$(\tau_1, \tau_2)$	$\Longrightarrow$	$(\tau'_1 \text{ lprod } \tau'_2)$
$[\tau]$	$\Longrightarrow$	$\tau' \text{ llist}$
$T \tau_1 \dots \tau_n$	$\Longrightarrow$	$\tau'_1 \dots \tau'_n T'$
		with $T$ either datatype or defined type

Built-in types are translated to the lifting of the corresponding HOL type. The Isabelle/HOLCF type constructor *lift* is used to lift types to flat domains. The type constructor *llist* is discussed in the next section.

## 4 Translation of Datatypes

As explained in the Haskell report [PJ03], section 4.2.3, the following four Haskell declarations

```
data D1 = D1 Int
data D2 = D2 !Int
type S = Int
newtype N = N Int
```

have four different semantics. Indeed, the correct translation to Isabelle/HOLCF is as follows:

```
domain D1 = D1 (lazy D1_1::"Int \mathit{lift}")
domain D2 = D2 (D2_1::"Int \mathit{lift}")
types S = "Int \mathit{lift}"
pcpodef N = "{x:: Int \mathit{lift} . True}"
by auto
```

In Isabelle/HOLCF, the keyword *domain* defines a (possibly recursive) domain as solution of the corresponding domain equation. The keyword *lazy* ensures that the constructor  $D1$  is non-strict, i.e.  $D1 \perp \neq \perp$ . The keyword *pcpodef* can be used to define sub-pcpo's of existing pcpo's; here, we use it just to introduce an isomorphic copy of an existing pcpo — this is the semantic of Haskell *newtype* definitions.

Lists are translated to the domain *llist*, defined as follows in our prelude theory *HsHOLCF*:

```
domain 'a llist = nil | ## (lazy HD :: 'a) (lazy TL :: 'a llist)
```

allowing for partial list as well as for infinite ones [MNvOS99].

For each datatype, we have to lift the constructors from the HOLCF continuous function spaces to our lifted function spaces:

```
constdefs
  cont2lifted2 :: "('a -> 'b -> 'c) => ('a --> 'b --> 'c)"
  "cont2lifted2 f == Lam x . Lam y. f $ x $ y"

  llCons :: "'a --> 'a llist --> 'a llist"
  "llCons == cont2lifted2 (op ##)"
```

The general scheme for translation of mutually recursive lazy Haskell datatypes to Isabelle/HOLCF domains is as follows:

$$\begin{aligned}
 \text{data } \phi_n &= C_{n1} w_1 \dots w_h \mid \dots \mid C_{1q} z_1 \dots z_k \implies \\
 \text{domain } \phi'_1 &= C'_{11} (\text{lazy} :: x'_1) \dots (\text{lazy} :: x'_i) \mid \\
 &\quad \dots \mid \\
 &\quad C'_{1p} (\text{lazy} :: y'_1) \dots (\text{lazy} :: y'_j) \\
 \text{and } \dots & \\
 \text{and } \phi'_n &= C'_{n1} (\text{lazy} :: w'_1) \dots (\text{lazy} :: w'_h) \mid \\
 &\quad \dots \mid \\
 &\quad C'_{nq} (\text{lazy} :: z'_1) \dots (\text{lazy} :: z'_k)
 \end{aligned}$$

Mutually recursive datatypes relies on specific Isabelle syntax (keyword *and*).<sup>3</sup> Order of declarations is taken care of. In case of strict arguments (indicated with ! in Haskell), the keyword *lazy* is omitted.

The translation scheme for type synonyms simply is

$$\text{type } \tau = \tau_1 \quad \implies \text{types } \tau = \tau'_1$$

## 5 Translation of Kinds and Type Classes

Type classes in Isabelle and Haskell associate a set of functions to a type class identifier (these are also called the *methods* of the class). In Isabelle, type classes are typically further specified using a set of axioms; for example, the class *linorder* of total orders is specified using the usual total order axioms. Of course, such axiomatizations are not possible in Haskell. Indeed, in Haskell, there is no check that the class *Ord* actually consists of total orders only, and hence it would be inappropriate to translate it to *linorder* in Isabelle. Instead, we translate to a newly declared Isabelle class *Ord*. The only thing that we assume is that it is a subclass of *pcpo*, because all Haskell types are translated to *pcpos*. Hence,

<sup>3</sup> Due to a bug in Isabelle/HOLCF 2005, declaration of mutually recursive lazy domains does not work. However, this will be fixed in Isabelle 2007; the fix currently is available in the development snapshot.

type classes are translated to Isabelle/HOLCF as subclasses of *pcpo* with empty axiomatization. Methods declarations associated with Haskell classes are translated to independent function declarations with appropriate class annotation on type variables.

Class instance declarations declare that a particular type belongs to a class. In Isabelle, instance declarations generate proof obligations, namely that the methods for the type at hand indeed satisfy the axioms of the class. Since our translations only generates classes without axioms (beyond those of *pcpo*), proofs are trivial and proof obligation may be automatically discharged.

A Haskell class instance declaration that declares type *T* to belong to class *C* may define the behaviour of the class methods for *T*. The same is possible with normal Isabelle constant definitions, if the type of the constant (function) is specialised to *T* in the definition. With this, we avoid an explicit handling of dictionaries, as described in the static semantics of Haskell [Fax02].

A restriction of Isabelle is that it does neither allow for multi-parameter classes, nor for type constructor ones. Therefore, the same restrictions apply to our translation.

Classes

$$K \Longrightarrow K'$$

Type schemas

$$\begin{array}{ccc} (\{K\ v\} \cup ctx) \Rightarrow \tau & \Longrightarrow & (ctx \Rightarrow \tau)' [(v' :: s)/(v' :: (K' \cup s))] \\ \{\} \Rightarrow \tau & \Longrightarrow & \tau' \end{array}$$

Haskell type variables are translated to variables of class *pcpo*. Each type is associated to a sort in Isabelle (in Haskell, the same concept is called “kind”), defined by the set of the classes of which it is member.

Class declarations

$$class\ K\ where\ (Dec_1; \dots; Dec_n) \Longrightarrow class\ K' \subseteq\ pcpo; Dec'_1; \dots; Dec'_n$$

Class instance definitions

$$\begin{array}{l} instance\ ctx \Rightarrow K_T\ (T\ v_1 \dots v_n)\ where \\ \quad (f_1 :: \tau_1 = t_1; \dots; f_n :: \tau_n = t_n) \Longrightarrow \\ instance \\ \quad \tau' :: K'_T\ (\{pcpo\} \cup \{K' : (K\ v_1) \in ctx\}, \dots, \\ \quad \quad \{pcpo\} \cup \{K' : (K\ v_n) \in ctx\}) \\ \quad with\ proof\ obligation; \\ defs\ f'_1 :: (ctx \Rightarrow \tau_1)' == t'_1; \dots; f'_n :: (ctx \Rightarrow \tau_n)' == t'_n \end{array}$$

## 6 Translation of Function Definitions and Terms

Terms of built-in type are translated using Isabelle/HOLCF-defined lifting function *Def*. The bottom element  $\perp$  is used for undefined terms. Isabelle/HOLCF-defined *lift1*  $:: ('a \Rightarrow' b :: pcpo) \Rightarrow ('a \text{ lift} \rightarrow' b)$  and *lift2*  $:: ('a \Rightarrow' b) \Rightarrow ('a \text{ lift} \rightarrow' b \text{ lift})$  are used to lift operators, as well as the following, defined in *HsHOLCF*.

$$\begin{aligned} \text{liftbin} &:: ('a \Rightarrow 'b \Rightarrow 'c) \Rightarrow ('a \text{ lift} \rightarrow 'b \text{ lift} \rightarrow 'c \text{ lift}) \\ \text{liftbin } f &== \text{lift1 } (\lambda x. \text{lift2 } (f x)) \end{aligned}$$

Boolean values are translated to values of *bool lift* (*tr* in Isabelle/HOLCF) i.e. *TT*, *FF* and  $\perp$ , and Boolean connectives to the corresponding Isabelle/HOLCF operators. Isabelle/HOLCF-defined *If then else fi* and *case* syntax are used to translate conditional and case expressions, respectively. There are restrictions, however, on case expressions, due to limitations in the translation of patterns; in particular, only simple patterns are allowed (no nested ones). On the other hand, Isabelle sensitiveness to the order of patterns in case expressions is dealt with. Multiple function definitions are translated as definitions based on case expressions. In function definitions as well as in case expressions, both wildcards — not available in Isabelle — and incomplete patterns — not allowed — are dealt with by elimination,  $\perp$  being used as default value in the latters. Only let expressions without patterns on the left are dealt with; where expressions, guarded expressions and list comprehension are not covered.

$$\begin{aligned} f &:: \phi && \Longrightarrow \text{consts } f' &:: \phi' \\ f \bar{x} p_1 \bar{x}_1 = t_1; \dots; f \bar{x} p_n \bar{x}_n = t_n &\Longrightarrow \\ &(f \bar{x} = \text{case } y \text{ of } (p_1 \rightarrow (\bar{x}_1 \rightarrow t_1); \dots; p_n \rightarrow (\bar{x}_n \rightarrow t_n)))' \\ f \bar{x} = t &\Longrightarrow \text{defs } f' &:: \phi' == \text{Lam } \bar{x}'. t' \\ &\text{with } f &:: \phi \text{ not occurring in } t \\ (f_1 \bar{v}_1 = t_1; \dots; f_n \bar{v}_n = t_n) &\Longrightarrow \\ \text{fixrec } f'_1 &:: \phi'_1 = (\text{Lam } \bar{v}'_1. t'_1) \text{ and} \\ &\dots \\ &\text{and } f'_n &:: \phi'_n = (\text{Lam } \bar{v}'_n. t'_n) \\ &\text{with } f_1 &:: \phi_1, \dots, f_n &:: \phi_n \text{ mutually recursive} \end{aligned}$$

Function declarations use Isabelle keyword *consts*. Non-recursive definitions are translated to standard definitions using Isabelle keyword *defs*. Recursive definitions rely on Isabelle/HOLCF package *fixrec* which provides nice syntax for fixed point definitions, including mutual recursion. Lambda abstraction is translated as continuous abstraction for lifted function spaces (*Lam*), function application as continuous application (the  $\$\$$  operator), see Sect. 3 above.

## 7 Example Proofs

The Haskell declarations

```
comp :: (b -> c) -> (a -> b) -> a -> c
comp f g x = f (g x)
```

```
map1 :: (a -> b) -> [a] -> [b]
map1 f [] = []
map1 f (x:xs) = f x : map1 f xs
```

are translated to

```
consts
X_comp :: "('b --> 'c) --> ('a --> 'b) --> 'a --> 'c"
map1 :: "('a --> 'b) --> 'a llist --> 'b llist"

defs
X_comp_def :
"(X_comp :: ('b --> 'c) --> ('a --> 'b) --> 'a --> 'c) ==
 Lam f. Lam g. Lam x. f $$ (g $$ x)"

fixrec "(map1 :: ('a --> 'b) --> 'a llist --> 'b llist) =
 (Lam qX1. (Lam qX2. case qX2 of
  lNil => lNil |
  op ### $ pX1 $ pX2 =>
    llCons $$ (qX1 $$ pX1) $$ (map1 $$ qX1 $$ pX2))))"
```

The fixrec definition leads to a bunch of theorems, one for simplification. However, the latter makes the Isabelle simplifier loop, and hence needs to be replaced with a more suitable simplification theorem:

```
lemmas map1_simps [simp del]
lemma map1_simp : " map1 $$ (f::('a --> 'b)) $$ (x::'a llist) =
  (case x of lNil => lNil |
   op ### $ pX1 $ pX2 =>
     llCons $$ (f$$pX1)$$ (map1$$f$$pX2))"
  apply (subst map1_simps)
  by auto
```

Due to the recursion, also this theorem cannot be fed into the simplifier. However, using substitution, it helps in proving the following expected lemmas about the behaviour of *map1*:

```
lemma map1_UU [simp] : "map1 $$ f $$ UU = UU"
  apply (subst map1_simp)
  by simp
```



```

lemma map1_lNil[simp] : " map1 $$ f $$ lNil = lNil"
  apply (subst map1_simp)
  by simp

lemma map1_cons1[simp] : "map1 $$ f $$ (op ### $ x $ (xs::'a llist)) =
  op ### $ (f $$ x) $ (map1 $$ f $$ xs)"
  apply (subst map1_simp)
  apply (simp add: cont2lifted2_def)
done

```

With these, it is now easy to prove that map distributes over composition via induction:

```

theorem compMap :
  "X_comp $$ (map1 $$ f) $$ (map1 $$ g) $$ x =
  map1 $$ (X_comp $$ f $$ g) $$ x"
  apply (rule llist.ind)
  back
  apply (auto simp add: X_comp_def)
done

```

## 8 Conclusion and Related Work

We provide a shallow embedding of Haskell to Isabelle/HOLCF, which can be used for proving properties of Haskell programs, but also is the first denotational semantics that has been given to Haskell.

The main advantage of our shallow approach is to get as much as possible out of the automation currently available in Isabelle, especially with respect to type checking. Isabelle/HOLCF in particular provides with an expressive semantics covering lazy evaluation, as well as with a smart syntax — also thanks to the *fixrec* package. It is interesting to note that Haskell functions and product types have to be lifted due to the mixture of eager and lazy evaluation that Haskell exhibits due to the presence of the *seq* function. Type classes in Haskell are similar enough to Isabelle's type classes such that explicit handling of dictionaries can be avoided.

The main disadvantage of our approach lies with lack of type constructor classes. Anyway, it is possible to get around the obstacle, at least partially, by relying on an axiomatic characterisation of monads and on a proof-reuse strategy that actually minimises the need for interactive proofs.

Concerning related work, although there have been translations of functional languages to first-order systems — those to FOL of Miranda [Tho94,Tho89,HT95] and Haskell [Tho92], both based on large-step operational semantics; that of Haskell to Agda implementation of Martin-Loef type theory in [ABB<sup>+</sup>05] — still, higher-order logic may be quite helpful in order to deal with features such as currying and polymorphism. Moreover, higher-order approaches may rely on denotational semantics — as for examples, [HMW05] translating

Haskell to HOLCF, and [LP04] translating ML to HOL — allowing for program representation closer to specification as well as for proofs comparatively more abstract and general.

The translation of Haskell to Isabelle/HOLCF proposed in [HMW05] uses deep embedding to deal with types. Haskell types are translated to terms, relying on a domain-theoretic modelling of the type system at the object level, allowing explicitly for a clear semantics, and providing for an implementation that can capture most features, including type constructor classes. In contrast, we provide in the case of Isabelle/HOLCF with a translation that follows the lines of a denotational semantics under the assumption that type constructors and type application in Haskell can be mapped to corresponding constructors and built-in application in Isabelle without loss from the point of view of behavioural equivalence between programs — in particular, translating Haskell datatypes to Isabelle ones. Our solution gives in general less expressiveness than the deeper approach — however, when we can get it to deal with cases of interest, it might make proofs easier.

operator.

Future work should use this framework for proving properties of Haskell programs. Currently Hets already provides some experimental translation of constructor classes and monads, also covering *do* notation, using theory morphisms as provided by the package AWE [BJL06]. However, there are mainly syntactic problems with name clashes that currently prevent a proper integration with Isabelle/HOLCF. These problems should be solved in the near future.

For monadic programs, we are also planning to use the monad-based dynamic Hoare and dynamic logic that already have been formalised in Isabelle [Wal05]. Our translation tool from Haskell to Isabelle is part of the Heterogeneous Tool Set Hets and can be downloaded from <http://www.dfki.de/sks/hets>. More details about the translations can be found in [TLMM07].

## Acknowledgment

This work has been supported by the *Deutsche Forschungsgemeinschaft* under grants KR 1191/5-2 and KR 1191/7-2 . We thank Brian Huffmann for help with the Isabelle/HOLCF package and Erwin R. Catesbeiana for pointing out an inconsistency.

## References

- [ABB<sup>+</sup>05] A. Abel, M. Benke, A. Bove, J. Hughes, and U. Norell. Verifying Haskell programs using constructive type theory. In *ACM-SIGPLAN 05*, 2005.
- [BJL06] M. Bortin, E. B. Johnsen, and C. Lueth. Structured formal development in Isabelle. *Nordic Journal of Computing*, 2006.
- [Fax02] Karl-Filip Faxén. A static semantics for haskell. *J. Funct. Program*, 12(4&5):295–357, 2002.

- [HCNP03] Paul Hudak, Antony Courtney, Henrik Nilsson, and John Peterson. Arrows, robots, and functional reactive programming. In *Summer School on Advanced Functional Programming 2002, Oxford University*, volume 2638 of *Lecture Notes in Computer Science*, pages 159–187. Springer-Verlag, 2003.
- [HH92] Kevin Hammond and Cordelia Hall. A dynamic semantics for haskell, October 20 1992.
- [HHJK04] T. Hallgren, J. Hook, M. P. Jones, and D. Kieburtz. An overview of the Programatica toolset. In *HCSS04*, 2004.
- [HMW05] B. Huffman, J. Matthews, and P. White. Axiomatic constructor classes in Isabelle-HOLCF. Research paper, OGI, 2005.
- [HT95] S. Hill and S. Thompson. Miranda in Isabelle. In *Proceedings of the first Isabelle users workshop*, number 397 in Technical Report, pages 122–135. University of Cambridge Computer Laboratory, 1995.
- [LP04] J. Longley and R. Pollack. Reasoning about CBV programs in Isabelle-HOL. In *TPHOL 04*, number 3223 in LNCS, pages 201–216. Springer, 2004.
- [MML07] Till Mossakowski, Christian Maeder, and Klaus Lüttich. The Heterogeneous Tool Set. In Orna Grumberg and Michael Huth, editors, *TACAS 2007*, volume 4424 of *Lecture Notes in Computer Science*, pages 519–522. Springer-Verlag Heidelberg, 2007.
- [MNvOS99] O. Mueller, T. Nipkow, D. von Oheimb, and O. Slotosch. HOLCF = HOL + LCF. *Journal of Functional Programming*, 1999.
- [Mos05] T. Mossakowski. Heterogeneous specification and the heterogeneous tool set, Habilitation Thesis, 2005.
- [Mos06] T. Mossakowski. Hets user guide. Tutorial, Universitaet Bremen, 2006.
- [Pau94] L. C. Paulson. *Isabelle: a generic theorem prover*, volume 828. Springer, 1994.
- [PHH99] John Peterson, Greg Hager, and Paul Hudak. A language for declarative robotic programming. In *International Conference on Robotics and Automation*, 1999.
- [PJ03] S. Peyton Jones, editor. *Haskell 98 Language and Libraries*. Cambridge University Press, 2003.
- [Tho89] S. Thompson. A logic for Miranda. *Formal Aspects of Computing*, 1, 1989.
- [Tho92] S. Thompson. Formulating Haskell. In *Functional Programming*. Springer, 1992.
- [Tho94] S. Thompson. A logic for Miranda, revisited. *Formal Aspects of Computing*, 3, 1994.
- [TLMM07] P. Torrini, C. Lueth, C. Maeder, and T. Mossakowski. Translating Haskell to Isabelle. Technical report, Universitaet Bremen, 2007.
- [Wal05] Dennis Walter. Monadic dynamic logic: Application and implementation, 2005.
- [Wen05] M. Wenzel. Using axiomatic type classes in Isabelle. Tutorial, TU Muenchen, 2005.
- [Win93] G. Winskel. *The Formal Semantics of Programming Languages*. MIT Press, 1993.